Pensieve header: A first implementation of nilpotent integration.

Initialization

```
In[∘]:= SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\HigherRank"];
        Once[<< KnotTheory`; << Rot.m];
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
Loading Rot.m from http://drorbn.net/icbs24 to compute rotation numbers.
```

```
In[∘]:= CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
        CCF[ℰ_] := Factor[ℰ];
        CF[ℰ_𝔼] := CF /@ ℰ;
        CF[ℰ_List] := CF /@ ℰ;
        CF[sd_SeriesData] := MapAt[CF, sd, 3];
        CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p)__, ∞] ⋃ {x, p}, ps, c},
          Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) ⧴ CCF[c] (Times @@ vs^ps)] ];
```

Integration

```
In[∘]:= 𝔼 /: 𝔼[A_] × 𝔼[B_] := 𝔼[A + B]
```

```
In[∘]:= Unprotect[Integrate];
        ∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, Q, G, V, s, t, k, a, b},
          n = Length@vs;
          Q = -Table[(∂_{vs⟦a⟧,vs⟦b⟧} L) /. Thread[vs → 0], {a, n}, {b, n}];
          G = Inverse[Q] / 2;
          V = L + vs.Q.vs / 2;
          s = t = V; k = 0;
          While[0 =!= t,
           s += 1/(++k)! (t = CF@Sum[G⟦a, b⟧ ((∂_{vs⟦a⟧,vs⟦b⟧} t) + (∂_{vs⟦a⟧} t) (∂_{vs⟦b⟧} t)), {a, n}, {b, n}])];
          PowerExpand@Factor[ω (Det[Q] (2 π)^n)^(-1/2)] × 𝔼[CF@s /. Thread[vs → 0]]
         ];
        Protect[Integrate];
```

In[∘]:= $\int \mathbb{E}\left[i\,\lambda\,x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$\frac{(-1)^{1/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-i\,\lambda\,x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$-\frac{(-1)^{3/4}\,\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{2\sqrt{b^2 - a\,c}\;\pi}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda\,x_1^2/2\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}[0]}{\sqrt{2\pi}\,\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[-x_1^2/2 + \xi\,x_1\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{\xi^2}{2}\right]}{\sqrt{2\pi}}$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\,(-b^2 + a\,c)}\right]}{2\sqrt{-b^2 + a\,c}\;\pi}$$

In[∘]:= $I1 = \int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-\frac{(-b^2 + a\,c)\,x_2^2}{2\,a} + \frac{\xi_1^2}{2\,a} + \frac{x_2\,(-b\,\xi_1 + a\,\xi_2)}{a}\right]}{\sqrt{a}\,\sqrt{2\pi}}$$

In[∘]:=
$$\int \mathtt{I1}\, \mathbb{d}\{x_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\xi_1^2-2\,b\,\xi_1\,\xi_2+a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:=
$$\int \mathbb{E}\left[-\frac{1}{2}\{y_1,y_2\}.\begin{pmatrix}a & b\\ b & c\end{pmatrix}.\{y_1,y_2\}+\{\eta_1,\eta_2\}.\{y_1,y_2\}\right]\mathbb{d}\{y_1,y_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{c\,\eta_1^2-2\,b\,\eta_1\,\eta_2+a\,\eta_2^2}{2\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:=
$$\mathtt{I1}=\int \mathbb{E}\left[-\frac{1}{2}\{y_1,y_2\}.\begin{pmatrix}a & b\\ b & c\end{pmatrix}.\{y_1,y_2\}+\{\eta_1,\eta_2\}.\{y_1,y_2\}\right]\mathbb{d}\{y_1\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{b^2\,y_2^2-a\,c\,y_2^2-2\,b\,y_2\,\eta_1+\eta_1^2+2\,a\,y_2\,\eta_2}{2\,a}\right]}{\sqrt{a}\;\sqrt{2\,\pi}}$$

In[∘]:=
$$\int \mathtt{I1}\, \mathbb{d}\{y_2\}$$

Out[∘]=
$$\frac{\mathbb{E}\left[\frac{a\,c\,\eta_1^2-2\,a\,b\,\eta_1\,\eta_2+a^2\,\eta_2^2}{2\,a\,(-b^2+a\,c)}\right]}{2\,\sqrt{-b^2+a\,c}\;\pi}$$

In[∘]:=
$$\int \mathbb{E}\left[\xi\,x+\eta\,y+z\,(x-y)+x^2\right]\mathbb{d}\{x,z\}$$

Out[∘]=
$$-\frac{i\,\mathbb{E}[y\,(y+\eta+\xi)]}{2\,\pi}$$

The ρ_1 Integrand

Adopted from pensieve://Projects/APAI/PerturbedGaussianIntegration.nb.

In[∘]:=
```
q[s_, i_, j_] := x_i (p_i - T^s p_{i+1} + (T^s - 1) p_{j+1}) + x_j (p_j - p_{j+1});
r_1[s_, i_, j_] :=
  s (-1 + 2 p_i x_i - 2 p_j x_i + (T^s - 1) p_i p_j x_i^2 + (1 - T^s) p_j^2 x_i^2 - 2 p_i p_j x_i x_j + 2 p_j^2 x_i x_j) / 2;
γ_1[φ_, k_] := ϵ φ (1 / 2 - x_k p_k);
ρ1i[s_, i_, j_] := T^(-s/2) 𝔼[-q[s, i, j] + ϵ r_1[s, i, j] + O[ϵ]^2];
ρ1i[φ_, k_] := T^(-φ/2) 𝔼[-x_k (p_k - p_{k+1}) + γ_1[φ, k] + O[ϵ]^2];
ρ1i[End, k_] := 𝔼[-x_k p_k + O[ϵ]^2];
ρ1i[K_] := Module[{Cs, φ, n, c, k, ℰ},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   ℰ = ρ1i[End, 2 n + 1];
   Do[ℰ *= ρ1i @@ c, {c, Cs}];
   Do[ℰ *= ρ1i[φ⟦k⟧, k], {k, 2 n}];
   CF@ℰ
  ];
ρ1vs[K_] := Union @@ Table[{x_i, p_i}, {i, 2 Crossings[K] + 1}]
```

In[∘]:=
```
ρ1i[Knot[3, 1]]
```

Out[∘]=

$$T^2\,\mathbb{E}\Big[\Big(-2\,(p_1-p_2)\,x_1-(p_2-p_3)\,x_2-\Big(p_2-\frac{p_3}{T}+\Big(-1+\frac{1}{T}\Big)\,p_6\Big)\,x_2-2\,(p_3-p_4)\,x_3-(p_4-p_5)\,x_4-\Big(\Big(-1+\frac{1}{T}\Big)\,p_2+p_4-\frac{p_5}{T}\Big)\,x_4-2\,(p_5-p_6)\,x_5-(p_6-p_7)\,x_6-\Big(\Big(-1+\frac{1}{T}\Big)\,p_4+p_6-\frac{p_7}{T}\Big)\,x_6-p_7\,x_7\Big)+\Big(-\frac{1}{2}+p_4\,x_4+\frac{1}{2}\Big(1+2\,p_1\,x_4-2\,p_4\,x_4-2\,p_1^2\,x_1\,x_4+2\,p_1\,p_4\,x_1\,x_4-\Big(1-\frac{1}{T}\Big)\,p_1^2\,x_4^2-\Big(-1+\frac{1}{T}\Big)\,p_1\,p_4\,x_4^2\Big)+\frac{1}{2}\Big(1-2\,p_2\,x_2+2\,p_5\,x_2-\Big(-1+\frac{1}{T}\Big)\,p_2\,p_5\,x_2^2-\Big(1-\frac{1}{T}\Big)\,p_5^2\,x_2^2+2\,p_2\,p_5\,x_2\,x_5-2\,p_5^2\,x_2\,x_5\Big)+\frac{1}{2}\Big(1+2\,p_3\,x_6-2\,p_6\,x_6-2\,p_3^2\,x_3\,x_6+2\,p_3\,p_6\,x_3\,x_6-\Big(1-\frac{1}{T}\Big)\,p_3^2\,x_6^2-\Big(-1+\frac{1}{T}\Big)\,p_3\,p_6\,x_6^2\Big)\Big)\,\epsilon+O[\epsilon]^2\Big]$$

In[∘]:=
```
ρ1vs[Knot[3, 1]]
```

Out[∘]=

$$\{p_1, p_2, p_3, p_4, p_5, p_6, p_7, x_1, x_2, x_3, x_4, x_5, x_6, x_7\}$$

Integration of ϵ-Series

In[*]:=
```
Unprotect[Integrate];
∫ω_. 𝔼[L_SeriesData] ⅆ(vs_List) := Module[{n, m, ϵ, L0, Q, Δ, G, V, Z, e, λ, k, a, b},
   ϵ = L[[1]]; m = L[[5]];
   n = Length@vs; L0 = Normal@L /. ϵ → 0;
   Q = -Table[(∂_{vs[[a]],vs[[b]]} L0) /. Thread[vs → 0] /. (p | x)__ → 0, {a, n}, {b, n}];
   If[(Δ = CF@Det[Q]) == 0,
    Return["How dare you ask me to integrate a singular Gaussian!"]];
   G = Inverse[Q];
   V = L + vs.Q.vs / 2;
   Z = V; k = 0;
   While[
    e = Normal@CF[(∂_λ Z) -
        1/2 Sum[G[[a, b]] ((∂_{vs[[a]],vs[[b]]} Z) + (∂_{vs[[a]]} Z) (∂_{vs[[b]]} Z)) + O[ϵ]^m, {a, n}, {b, n}]];
    0 =!= e,
    Z -= λ^(k+1)/(k + 1) Coefficient[e, λ, k];
    ++k
   ];
   PowerExpand@Factor[ω (Δ (2 π)^n)^(-1/2)] × 𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]
  ];
Protect[Integrate];
```

In[*]:=
```
∫𝔼[-x^2 / 2 + ϵ x^3 / 6 + O[ϵ]^20] ⅆ{x}
```

Out[*]=

$$\frac{1}{\sqrt{2\pi}}\,\mathbb{E}\Big[\frac{5\epsilon^2}{24}+\frac{5\epsilon^4}{16}+\frac{1105\epsilon^6}{1152}+\frac{565\epsilon^8}{128}+\frac{82825\epsilon^{10}}{3072}+\frac{19675\epsilon^{12}}{96}+\frac{1282031525\epsilon^{14}}{688128}+\frac{80727925\epsilon^{16}}{4096}+\frac{1683480621875\epsilon^{18}}{7077888}+O[\epsilon]^{20}\Big]$$

In[∘]:= $\int \mathbb{E}\left[-x^2/2 + \epsilon\, x^4/24 + O[\epsilon]^{16}\right] \mathbb{d}\{x\}$

Out[∘]=

$$\frac{1}{\sqrt{2\pi}}\,\mathbb{E}\Big[\frac{\epsilon}{8} + \frac{\epsilon^2}{12} + \frac{11\,\epsilon^3}{96} + \frac{17\,\epsilon^4}{72} + \frac{619\,\epsilon^5}{960} + \frac{709\,\epsilon^6}{324} + \frac{858\,437\,\epsilon^7}{96\,768} + \frac{54\,193\,\epsilon^8}{1296} + \frac{18\,639\,247\,\epsilon^9}{82\,944} + \frac{2\,197\,187\,\epsilon^{10}}{1620} + \frac{33\,152\,545\,703\,\epsilon^{11}}{3\,649\,536} + \frac{1\,169\,890\,097\,\epsilon^{12}}{17\,496} + \frac{41\,657\,327\,595\,361\,\epsilon^{13}}{77\,635\,584} + \frac{31\,722\,037\,141\,\epsilon^{14}}{6804} + \frac{6\,944\,870\,083\,473\,751\,\epsilon^{15}}{159\,252\,480} + O[\epsilon]^{16}\Big]$$

In[∘]:= $\int \mathbb{E}\left[x_1 p_1 + \epsilon\, x_1^7 p_1^7 + O[\epsilon]^2\right] \mathbb{d}\{x_1, p_1\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-5040\,\epsilon + O[\epsilon]^2\right]}{2\pi}$$

In[∘]:= $\int \mathbb{E}\left[x_1 p_2 + \epsilon\, x_2^7 p_1^7 + O[\epsilon]^2\right] \mathbb{d}\{x_1, p_2\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[p_1^7 x_2^7\,\epsilon + O[\epsilon]^2\right]}{2\pi}$$

In[∘]:= $\int \mathbb{E}\left[x_1 p_2 + 3\, x_2 p_1 + \epsilon\, p_2^5 x_1^5 + O[\epsilon]^2\right] \mathbb{d}\{x_1, x_2, p_1, p_2\}$

Out[∘]=

$$\frac{\mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{12\pi^2}$$

In[∘]:= $\int \mathbb{E}\left[x_1 p_2 + x_2 p_3 + x_3 p_1 + \epsilon\, x_1^5 p_2^5 + O[\epsilon]^2\right] \mathbb{d}\{x_1, x_2, x_3, p_1, p_2, p_3\}$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[-120\,\epsilon + O[\epsilon]^2\right]}{8\pi^3}$$

In[∘]:= MatrixForm@Table[
$\int \mathbb{E}[x_1 p_2 + x_2 p_3 + x_3 p_1 + \xi_i x_i + \pi_j p_j]\, \mathbb{d}\{x_1, x_2, x_3, p_1, p_2, p_3\}$,
{i, 3}, {j, 3}]

Out[∘]//MatrixForm=

$$\begin{pmatrix} -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_2 \xi_1]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[-\pi_3 \xi_2]}{8\pi^3} \\ -\frac{i\,\mathbb{E}[-\pi_1 \xi_3]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} & -\frac{i\,\mathbb{E}[0]}{8\pi^3} \end{pmatrix}$$

In[∘]:=
```
K = Knot[5, 2];
{ρ1i@K, ρ1vs@K}
∫ρ1i[K] d(ρ1vs@K)
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]=

$$\Big\{T^3\,\mathbb{E}\Big[\Big(-2\,(p_1-p_2)\,x_1-(p_2-p_3)\,x_2-\Big(p_2-\frac{p_3}{T}+\Big(-1+\frac{1}{T}\Big)\,p_8\Big)\,x_2-2\,(p_3-p_4)\,x_3-(p_4-p_5)\,x_4-\Big(\Big(-1+\frac{1}{T}\Big)\,p_2+p_4-\frac{p_5}{T}\Big)\,x_4-2\,(p_5-p_6)\,x_5-(p_6-p_7)\,x_6-\Big(p_6-\frac{p_7}{T}+\Big(-1+\frac{1}{T}\Big)\,p_{10}\Big)\,x_6-2\,(p_7-p_8)\,x_7-(p_8-p_9)\,x_8-\Big(\Big(-1+\frac{1}{T}\Big)\,p_4+p_8-\frac{p_9}{T}\Big)\,x_8-2\,(p_9-p_{10})\,x_9-(p_{10}-p_{11})\,x_{10}-\Big(\Big(-1+\frac{1}{T}\Big)\,p_6+p_{10}-\frac{p_{11}}{T}\Big)\,x_{10}-p_{11}\,x_{11}\Big)+\Big(-\frac{1}{2}+p_4\,x_4+\frac{1}{2}\Big(1+2\,p_1\,x_4-2\,p_4\,x_4-2\,p_1^2\,x_1\,x_4+2\,p_1\,p_4\,x_1\,x_4-\Big(1-\frac{1}{T}\Big)\,p_1^2\,x_4^2-\Big(-1+\frac{1}{T}\Big)\,p_1\,p_4\,x_4^2\Big)+\frac{1}{2}\Big(1-2\,p_2\,x_2+2\,p_7\,x_2-\Big(-1+\frac{1}{T}\Big)\,p_2\,p_7\,x_2^2-\Big(1-\frac{1}{T}\Big)\,p_7^2\,x_2^2+2\,p_2\,p_7\,x_2\,x_7-2\,p_7^2\,x_2\,x_7\Big)+\frac{1}{2}\Big(1+2\,p_3\,x_8-2\,p_8\,x_8-2\,p_3^2\,x_3\,x_8+2\,p_3\,p_8\,x_3\,x_8-\Big(1-\frac{1}{T}\Big)\,p_3^2\,x_8^2-\Big(-1+\frac{1}{T}\Big)\,p_3\,p_8\,x_8^2\Big)-p_9\,x_9+\frac{1}{2}\Big(1-2\,p_6\,x_6+2\,p_9\,x_6-\Big(-1+\frac{1}{T}\Big)\,p_6\,p_9\,x_6^2-\Big(1-\frac{1}{T}\Big)\,p_9^2\,x_6^2+2\,p_6\,p_9\,x_6\,x_9-2\,p_9^2\,x_6\,x_9\Big)+p_{10}\,x_{10}+\frac{1}{2}\Big(1+2\,p_5\,x_{10}-2\,p_{10}\,x_{10}-2\,p_5^2\,x_5\,x_{10}+2\,p_5\,p_{10}\,x_5\,x_{10}-\Big(1-\frac{1}{T}\Big)\,p_5^2\,x_{10}^2-\Big(-1+\frac{1}{T}\Big)\,p_5\,p_{10}\,x_{10}^2\Big)\Big)\,\epsilon+O[\epsilon]^2\Big],\ \{p_1,p_2,p_3,p_4,p_5,p_6,p_7,p_8,p_9,p_{10},p_{11},x_1,x_2,x_3,x_4,x_5,x_6,x_7,x_8,x_9,x_{10},x_{11}\}\Big\}$$

Out[∘]=

$$-\frac{i\,T^7\,\mathbb{E}\Big[\frac{(1+6\,T+23\,T^2-38\,T^3+105\,T^4-88\,T^5+71\,T^6)\,\epsilon}{8\,T\,(1+T)\,(1-T+2\,T^2)^2}+O[\epsilon]^2\Big]}{262144\,\pi^{11}\,(1+T)^2\,(1-T+2\,T^2)}$$

In[∘]:=
```
K = Knot[8, 19];
{ρ1i@K, ρ1vs@K}
∫𝔼[ρ1i@K] d(ρ1vs@K)
```

Out[∘]=

$$\Big\{\frac{1}{T^4}\,\mathbb{E}\Big[(-((p_1-p_2)\,x_1)-(p_1-T\,p_2+(-1+T)\,p_5)\,x_1-2\,(p_2-p_3)\,x_2-(p_3-p_4)\,x_3-(p_3-T\,p_4+(-1+T)\,p_9)\,x_3-2\,(p_4-p_5)\,x_4-2\,(p_5-p_6)\,x_5-(p_6-p_7)\,x_6-(p_6-T\,p_7+(-1+T)\,p_{14})\,x_6-(p_7-p_8)\,x_7-((-1+T)\,p_3+p_7-T\,p_8)\,x_7-2\,(p_8-p_9)\,x_8-2\,(p_9-p_{10})\,x_9-(p_{10}-p_{11})\,x_{10}-(p_{10}-T\,p_{11}+(-1+T)\,p_{16})\,x_{10}-2\,(p_{11}-p_{12})\,x_{11}-(p_{12}-p_{13})\,x_{12}-((-1+T)\,p_6+p_{12}-T\,p_{13})\,x_{12}-2\,(p_{13}-p_{14})\,x_{13}-(p_{14}-p_{15})\,x_{14}-((-1+T)\,p_{10}+p_{14}-T\,p_{15})\,x_{14}-$$

$$2\,(p_{15}-p_{16})\,x_{15}-(p_{16}-p_{17})\,x_{16}-((-1+T)\,p_{12}+p_{16}-T\,p_{17})\,x_{16}-p_{17}\,x_{17})+$$
$$\Big(p_4\,x_4+\frac{1}{2}\,\big(-1+2\,p_1\,x_1-2\,p_4\,x_1+(-1+T)\,p_1\,p_4\,x_1^2+(1-T)\,p_4^2\,x_1^2-2\,p_1\,p_4\,x_1\,x_4+2\,p_4^2\,x_1\,x_4\big)+$$
$$\frac{1}{2}\,\big(-1-2\,p_2\,x_7+2\,p_7\,x_7+2\,p_2^2\,x_2\,x_7-2\,p_2\,p_7\,x_2\,x_7+(1-T)\,p_2^2\,x_7^2+(-1+T)\,p_2\,p_7\,x_7^2\big)+$$
$$\frac{1}{2}\,\big(-1+2\,p_3\,x_3-2\,p_8\,x_3+(-1+T)\,p_3\,p_8\,x_3^2+(1-T)\,p_8^2\,x_3^2-2\,p_3\,p_8\,x_3\,x_8+2\,p_8^2\,x_3\,x_8\big)-p_{12}\,x_{12}+$$
$$\frac{1}{2}\,\big(-1-2\,p_5\,x_{12}+2\,p_{12}\,x_{12}+2\,p_5^2\,x_5\,x_{12}-2\,p_5\,p_{12}\,x_5\,x_{12}+(1-T)\,p_5^2\,x_{12}^2+(-1+T)\,p_5\,p_{12}\,x_{12}^2\big)+$$
$$\frac{1}{2}\,\big(-1+2\,p_6\,x_6-2\,p_{13}\,x_6+(-1+T)\,p_6\,p_{13}\,x_6^2+(1-T)\,p_{13}^2\,x_6^2-2\,p_6\,p_{13}\,x_6\,x_{13}+2\,p_{13}^2\,x_6\,x_{13}\big)+$$
$$\frac{1}{2}\,\big(-1-2\,p_9\,x_{14}+2\,p_{14}\,x_{14}+2\,p_9^2\,x_9\,x_{14}-2\,p_9\,p_{14}\,x_9\,x_{14}+(1-T)\,p_9^2\,x_{14}^2+(-1+T)\,p_9\,p_{14}\,x_{14}^2\big)+$$
$$\frac{1}{2}\,\big(-1+2\,p_{10}\,x_{10}-2\,p_{15}\,x_{10}+(-1+T)\,p_{10}\,p_{15}\,x_{10}^2+(1-T)\,p_{15}^2\,x_{10}^2-$$
$$2\,p_{10}\,p_{15}\,x_{10}\,x_{15}+2\,p_{15}^2\,x_{10}\,x_{15}\big)+\frac{1}{2}\,\big(-1-2\,p_{11}\,x_{16}+2\,p_{16}\,x_{16}+2\,p_{11}^2\,x_{11}\,x_{16}-$$
$$2\,p_{11}\,p_{16}\,x_{11}\,x_{16}+(1-T)\,p_{11}^2\,x_{16}^2+(-1+T)\,p_{11}\,p_{16}\,x_{16}^2\big)\Big)\,\epsilon+O[\epsilon]^2\Big],$$

$\{p_1,\ p_2,\ p_3,\ p_4,\ p_5,\ p_6,\ p_7,\ p_8,\ p_9,\ p_{10},\ p_{11},\ p_{12},$
$p_{13},$
$p_{14},$
$p_{15},$
$p_{16},$
$p_{17},$
$x_1,$
$x_2,$
$x_3,$
$x_4,$
$x_5,$
$x_6,$
$x_7,$
$x_8,$
$x_9,$
$x_{10},$
$x_{11},$
$x_{12},$
$x_{13},$
$x_{14},$
$x_{15},$
$x_{16},$
$x_{17}\}\}$

Series: Division by a series with no coefficients in $\frac{1}{O[\epsilon]^2}$.

Series: Division by a series with no coefficients in $\frac{1}{O[\epsilon]^4}$.

Series: Division by a series with no coefficients in $\frac{1}{O[\epsilon]^6}$.

General: Further output of Series::sbyc will be suppressed during this calculation.

Out[∘]=

```
$Aborted
```

Invariance Under Reidemeister 3b

In[∘]:= $\text{lhs} = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \pi_k p_k + \mathcal{L}[1, i, j] + \mathcal{L}[1, i+1, k] + \mathcal{L}[1, j+1, k+1] + O[\epsilon]^2\right] \, d\{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}\}$

Out[∘]=

$$\frac{1}{64\pi^6}\mathbb{E}\left[\left(T^2 p_{2+i}\pi_i - T p_{2+j}(-\pi_i + T\pi_i - \pi_j) + p_{2+k}(\pi_i - T\pi_i + \pi_j - T\pi_j + \pi_k)\right) - \frac{3\epsilon}{2} + O[\epsilon]^2\right]$$

In[∘]:= $\text{rhs} = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \pi_k p_k + \mathcal{L}[1, j, k] + \mathcal{L}[1, i, k+1] + \mathcal{L}[1, i+1, j+1] + O[\epsilon]^2\right] \, d\{x_i, x_j, x_k, p_i, p_j, p_k, x_{i+1}, x_{j+1}, x_{k+1}, p_{i+1}, p_{j+1}, p_{k+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[\left(T^2 p_{2+i}\pi_i - T p_{2+j}(-\pi_i + T\pi_i - \pi_j) + p_{2+k}(\pi_i - T\pi_i + \pi_j - T\pi_j + \pi_k)\right) - \frac{3\epsilon}{2} + O[\epsilon]^2\right]}{64\pi^6}$$

In[∘]:= lhs == rhs

Out[∘]=

```
True
```

Invariance Under Reidemeister 2b

In[∘]:= $\text{lhs} = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \mathcal{L}[1, i, j] + \mathcal{L}[-1, i+1, j+1] + O[\epsilon]^2\right] \, d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\epsilon]^2\right]}{16\pi^4}$$

In[∘]:= $\text{rhs} = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \mathcal{L}[0, i] + \mathcal{L}[0, i+1] + \mathcal{L}[0, j] + \mathcal{L}[0, j+1] + O[\epsilon]^2\right] \, d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + O[\epsilon]^2\right]}{16\pi^4}$$

In[∘]:= lhs == rhs

Out[∘]=

True

Invariance Under R2c

In[∘]:= $lhs = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \mathcal{L}[-1, i, j+1] + \mathcal{L}[1, i+1, j] + \gamma_1[-1, j+1] + O[\epsilon]^2\right] d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2\right]}{16\pi^4}$$

In[∘]:= $rhs = \int \mathbb{E}\left[\pi_i p_i + \pi_j p_j + \mathcal{L}[0, i] + \mathcal{L}[0, i+1] + \mathcal{L}[0, j] + \mathcal{L}[-1, j+1] + O[\epsilon]^2\right] d\{x_i, x_j, p_i, p_j, x_{i+1}, x_{j+1}, p_{i+1}, p_{j+1}\}$

Out[∘]=

$$\frac{\mathbb{E}\left[(p_{2+i}\pi_i + p_{2+j}\pi_j) + \frac{\epsilon}{2} + O[\epsilon]^2\right]}{16\pi^4}$$

In[∘]:= lhs == rhs

Out[∘]=

True

Invariance Under R1l

In[∘]:= $lhs = \int \mathbb{E}\left[\pi_i p_i + \mathcal{L}[1, i+2, i] + \mathcal{L}[1, i+1] + O[\epsilon]^2\right] d\{x_i, x_{i+1}, x_{i+2}, p_i, p_{i+1}, p_{i+2}\}$

$$» \begin{pmatrix} 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & -1+T & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ -1 & 1 & -1+T & 0 & 0 & 0 \\ 0 & -1 & 1 & 0 & 0 & 0 \end{pmatrix}$$

Out[∘]=

$$-\frac{i\,\mathbb{E}\left[p_{3+i}\pi_i + O[\epsilon]^2\right]}{8\pi^3 T}$$

In[ ]:= **rhs** = $\int \mathbb{E}\left[\pi_i\, p_i + \mathcal{L}[0, i] + \mathcal{L}[0, i+1] + \mathcal{L}[0, i+2] + O[\epsilon]^2\right] d\{x_i, x_{i+1}, x_{i+2}, p_i, p_{i+1}, p_{i+2}\}$

» $$\begin{pmatrix} 0 & 0 & 0 & 1 & -1 & 0 \\ 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 & 0 & 0 \\ 0 & -1 & 1 & 0 & 0 & 0 \end{pmatrix}$$

Out[ ]=

$$-\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3}$$

In[ ]:= **lhs == rhs**

Out[ ]=

$$-\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3\, T} == -\frac{i\, \mathbb{E}\left[p_{3+i}\, \pi_i + O[\epsilon]^2\right]}{8\, \pi^3}$$